

Paper 220-2013

## Propensity Score-based Analysis of Short-Term Complications in Patients with Lumbar Discectomy in the ACS-NSQIP Database

Yubo Gao The University of Iowa Hospitals and Clinics, Iowa City, Iowa

### ABSTRACT

Lumbar discectomy is the most common spinal procedure performed, and can be done on an outpatient basis. In this study, we want to compare the incidence of complications in patients undergoing single level lumbar discectomy between the inpatient and outpatient settings, to determine baseline 30-day complication rates, and to identify independent risk factors for complications. To achieve those, patients undergoing lumbar discectomy between 2005 and 2010 were selected from the ACS NSQIP database, based on a single primary CPT code. Thirty-day post-operative complications and pre-operative patient characteristics were identified and compared. Propensity score matching and multivariate logistic regression analysis were used to adjust for selection bias and identify predictors of 30-day morbidity. All analyses are performed via SAS ® software.

### INTRODUCTION

Lumbar discectomy is the most common spinal surgical procedure, with over 300,000 performed annually.<sup>1,2</sup> It can be done on either an outpatient or inpatient basis. The primary benefit of an inpatient stay is close monitoring of the surgical wound for signs of infection or hematoma formation, and for signs of neurologic deficit which might require urgent intervention. In contrast, outpatients avoid the nosocomial risks of hospitalization, and outpatient discectomies have been performed since at least 1985. Since that time, several studies have reported low complication rates in an outpatient setting, with satisfaction rates similar to that reported with inpatient hospitalization.<sup>1,3-5</sup> However, to the best of our knowledge, no large prospective study has ever directly compared the incidence of complications between inpatient and outpatients undergoing discectomy, and thus the relative safety the outpatient protocol remains unknown in this patient population.

The purposes of this study were to compare the incidence of complications in patients undergoing single level lumbar discectomy between the inpatient and outpatient settings, to determine baseline 30-day complication rates for lumbar discectomy, and to identify independent risk factors for complications. We queried a database prospectively maintained by the American College of Surgeons (ACS) National Surgical Quality Improvement Program (NSQIP)<sup>6</sup> to identify all lumbar discectomy cases performed between 2005 and 2010, and hypothesized that because of inherent exposure and care process differences, significant differences may exist between patients who underwent outpatient procedures as compared to those with inpatient hospitalization.

### METHODS

#### (1) DATA SELECTION

The American College of Surgeons (ACS) National Surgical Quality Improvement Program (NSQIP), known as ACS NSQIP, database provides surgical outcomes data for participating institutions. The use of the ACS NSQIP database has been widely accepted for use in short term surgical outcomes research in a variety of surgical specialties.<sup>7-10</sup> Patients who underwent a lumbar discectomy were identified from the ACS NSQIP database including the years of 2005 through 2010 using a single primary Current Procedural Terminology (CPT) code 63030. Detailed information on patient demographics, pre-operative comorbidities, laboratory values, and operative variables are collected. Post-operative 30-day outcomes across 21 categories of morbidity and mortality are recorded by approved surgical clinical reviewers. Data collection also continues across 30 days regardless of inpatient status. Finally, 4,310 cases were used, 1,652 (38.2%) of whom were treated as outpatients and 2,658 (61.8%) treated as inpatients.

#### (2) EXPLANATORY AND CONTROL VARIABLES

Patient demographics included: age, sex, and race (white, black, other). Preoperative health variables included body mass index (BMI), recent weight loss, diabetes mellitus, smoking, alcohol use, pre-operative blood transfusion, recent operation (within 30 days), and corticosteroid use. Preoperative comorbidities included coronary artery disease (chronic heart failure and/or myocardial infarction), peripheral vascular disease, previous transient ischemic attack

(TIA), dialysis use, bleeding disorder, open wound infection, chemotherapy, radiation therapy, pre-operative sepsis and chronic obstructive pulmonary disease (COPD), Preoperative laboratory values included: white blood cell count (WBC), hematocrit (HCT), platelet count, creatinine, albumin, and international normalized ratio (INR). The operative variables included: wound class, American Society of Anesthesiologists (ASA) class, number of blood transfusions, length of operation, and resident involvement.

### (3) OUTCOMES

Short term, 30-day, complications after spine surgery were: wound complications (superficial wound infection, deep wound infection, wound dehiscence, deep organ space infection); pulmonary complication (pneumonia and unplanned intubation); hematologic complication (pulmonary embolism and deep vein thrombosis); cardiac complication (acute myocardial infarction and cardiac arrest requiring resuscitation); renal complication (renal failure or insufficiency defined as a rise in Cr > 2mg/dL above baseline); neurologic complication (stroke, coma lasting greater than 24 hours, peripheral nerve injury); symptomatic urinary tract infection, bleeding requiring transfusion, graft failure, sepsis, septic shock, return to the operating room within 30 days, length of hospital stay (LOS) and mortality within 30 days. Any complication was defined by the presence of one or more of the above positive outcomes.

### (4) STATISTICAL ANALYSIS AND SAS<sup>11</sup> PROCEDURES

First, a simple unadjusted univariate comparison via Student two tailed t-tests and chi-square analysis was used to demonstrate differences for individual and combined complication rates where the SAS® codes (PROC TTEST and PROC FREQ) are not shown here due to simplicity. Second, in order to reduce selection bias, propensity score matching and adjusted multivariate logistic regression analysis were used to compare complications between the inpatient and outpatient cohorts while controlling for potential confounders and to identify predictors of 30-day morbidity. We chose to use two separate statistical models because a positive result from two separate statistical methods holds greater validity than a single method alone.

Independent variables for consideration of input into the propensity score and multivariate model included age (called age1 in the model, and so on), gender (sex), race (new\_race), BMI, recent weight loss (wtloss), diabetes mellitus (diabetes1), dialysis use (dialysis), open wound or current wound infection (wndinf), recent operation (proper30), white blood cell count (prwbc), hematocrit (prhct), wound class (wndclas), ASA class (asaclas), operation time (optime), and intra-op blood transfusions (rbc).

First, we build the propensity score model. The propensity score here is defined as the conditional probability of receiving inpatient treatment based on observed co-variants. In this study, the co-variants are the suspected independent comorbidities identified by the univariate analysis described above. The calculated propensity score for each patient, a numeric variable between 0 and 1, serves a single index of probability of receiving inpatient treatment. The codes are listed below.

```
/*propensity_scores*/
proc logistic data=pgy2 descending;
  class sex new_race wtloss diabetes1 dialysis wndinf proper30 wndclas asaclas;
  model inout2=sex new_race wtloss diabetes1 dialysis wndinf proper30 wndclas asaclas age1 bmi
    prwbc prhct optime rbc/lackfit risklimits;
  output out=propensity_scores pred=prob_inout;
run;
quit;
```

Here pgy2 is the study dataset where inout2=1 if inpatient, and 0 if outpatient. propensity\_scores is the output dataset containing both all the covariates and the propensity scores called prob\_inout.

Before we analyze further, we need compare their propensity score distributions that are typically referred to as the evaluation of common support. Propensity score distributions should overlap mostly indicating that observations from both groups are available across the range of the propensity scores. For this study, Figure 1 shows propensity score distributions, which overlap a lot and means the groups are comparable.

```
proc univariate data=propensity_scores noprint;
  class inout2;
  histogram prob_inout;
  inset N='N' (6.0)
  Mean='Mean' (4.2) STD='STD' (4.2)/font='Arial' pos=ne height=3;
run;
```




Figure 1: propensity score distributions

Then using following codes divide the propensity scores into five groups.

```
proc rank groups=5 data=propensity_scores out=r;
  ranks rnks;
  var prob_inout;
run;

data a;
  set r;
  quintile=rnks+1;
run;
```

Next, using quintile (or even directly using prob\_inout) adjusts selection bias in the following codes for complication comparison.

```
proc logistic data=a descending;
  class inout2;
  model complication= inout2 quintile/lackfit risklimits;
run;
quit;
```

Below are codes for comparing variables between two groups after quintile or propensity score adjustments.

```
/*quintile adjusted comparisons for continuous variables*/
proc glm data=a;
  class inout2 ;
  model continuous_variables=inout2 quintile/ss3 solution;
  lsmeans inout2/adjust=tukey pdiff cl;
quit;
```



```

set outpatients; /*Treatment*/ /*12*/
retain BestDistance 99; /*13*/
rc= iter.first(); /*14*/
if (rc=0) then BestDistance= 99; /*15*/
do while (rc= 0); /*16*/
    ScoreDistance= abs(pscoreT - pscoreC); /*17*/
    if ScoreDistance < BestDistance then do; /*18*/
        BestDistance= ScoreDistance; /*19*/
        IdSelectedControl= idC; /*20*/
        MatchedToTreatID= idT; /*21*/
    end; /*22*/
rc= iter.next(); /*23*/
if (rc~= 0) then do; /*24*/
    output; /*25*/
    rc1= h.remove(key: IdSelectedControl); /*26*/
end; /*27*/
end; /*28*/
run; /*29*/

```

The Control dataset inpatients is stored in a hash and then the hash is iterated. The syntax that is part of the definition of the hash object starts in line 2 and ends in line 11. The hash called "h" is created (declared) in line 5. Line 5 specifies that the hash h contains the Control dataset inpatients. Besides "declaring" a hash, the hash iterator object needs to be explicitly declared, which is accomplished in line 6. The iterator is called "iter." Line 7 and 8 explicitly define the key and the data. The key in this case is the variable idC since it will be used to identify the control observations, and the hash data (defined in line 8) is made of the variables pscoreC and idC. Note that the key idC is also part of the data since the contents of the variable idC are used not only as identifiers (in line 20). After declaring the hash, line 12 starts reading the Treatment dataset outpatients. The retain statement in line 13 is used to save the value of the best distance. In line 14, the variable rc equals 0 for the first observation of the hash h and it is used in line 15 to set the BestDistance variable to 99. The loop that iterates the hash h starts in line 16 and ends in line 28. The idea is to iterate the hash h in search of the best distance between the propensity scores. After all the observations in the Control dataset inpatients have been considered, the match information is saved by exporting the variables IdSelectedControl and MatchedToTreatID in line 25 by using the output statement. In other words, the information in the hash is passed to the dataset Matched. In the nearest available neighbor without replacement method, matched controls are not considered again once they have been matched. The method "remove" in line 26 is used to remove the observation identified by the key IdSelectedControl. When the next treated unit is read, the control unit matched previously is not part of the hash h.

```

/*select those matched inpatients*/
proc sql noprint;
    create table Tinpatients as
    select a.*,b.*
    from matched a join inpatients b
    on a.IdSelectedControl=b.idC;
quit;

proc freq data=Tinpatients;
    tables idC;
run;

proc freq data=Tinpatients;
    tables IdSelectedControl;
run;

data analysis;
    set outpatients Tinpatients;
run;

data aa;
    set analysis;
    prob=pscoreT;
    if pscoreC ne . and pscoreT=. then prob=pscoreC;
run;

```

We can further compare propensity score distributions using **proc univariate** or **proc boxplot** to see if selection bias is reduced.

```

title 'matched plot';
proc univariate data=aa noprint;
  class inout2;
  histogram prob;
  inset N='N' (6.0)
  Mean='Mean' (4.2) STD='STD' (4.2)/font='Arial' pos=ne height=3;
run;

```




Figure 2: propensity score distributions after matching

From Figure 1 & 2, it can be seen that matching indeed reduced bias.

After the matched pairs have been identified, the difference between the continuous outcome means in the two groups can be tested using PROC TTEST with PAIRED statement, see Lanehart<sup>13</sup> or the signed rank test, see Parsons<sup>14</sup>, while for categorical variables analysis using McNemar's test in PROC FREQ, see Parsons. The codes for those after-match comparisons are omitted due to simplicity.

## RESULTS

The study showed that age, diabetes, presence of pre-operative wound infection, blood transfusion, operative time exceeding 150 minutes, and an inpatient hospital stay were all independent risk factors of short term complication after lumbar discectomy. A multivariate logistic regression analysis produced similar results with inpatients having a higher risk of complications (Odds Ratio 1.521; 95% CI 1.048-2.206). After propensity score matching, overall complication rates were higher for inpatients as compared to outpatients (5.4% vs 3.5%,  $p = 0.0068$ ). For detailed results, please refer to<sup>15</sup>.

## CONCLUSIONS

Overall, this is the largest study so far to report on differences in complications between outpatient and inpatient lumbar discectomy procedures, including data collected from 258 hospitals around the United States. Thus, our results are highly generalizable, and can serve to inform patient discussion and serve as a standard against which to compare the complication rates of individual surgeons or institutions. We have shown that patients undergoing outpatient lumbar discectomy have lower peri-operative complication rates than those undergoing inpatient procedures. Increasing age, diabetes, pre-existing wound infection, blood transfusion, and longer operative times were independent risk factors for peri-operative complication, and caution should be used before pursuing outpatient discectomy in patients with one of those risk factors.

## REFERENCES

1. Best, NM., and Sasso, RC. Success and safety in outpatient microlumbar discectomy. *J Spinal Disord Tech*, 19(5): 334-7, 2006.
2. Sherman, J.; Cauthen, J.; Schoenberg, D.; Burns, M.; Reaven, NL.; and Griffith, SL. Economic impact of improving outcomes of lumbar discectomy. *Spine J*, 10(2): 108-16, 2010.
3. Zahrawi, F. Microlumbar discectomy. Is it safe as an outpatient procedure? *Spine (Phila Pa 1976)*, 19(9): 1070-4, 1994.
4. Singhal, A., and Bernstein, M. Outpatient lumbar microdiscectomy: a prospective study in 122 patients. *Can J Neurol Sci*, 29(3): 249-52, 2002.
5. Bookwalter, JW., 3rd; Busch, MD.; and Nicely, D. Ambulatory surgery is safe and effective in radicular disc disease. *Spine (Phila Pa 1976)*, 19(5): 526-30, 1994.
6. The American College of Surgeons (ACS) National Surgical Quality Improvement Program (NSQIP). American College of Surgeons, Chicago, IL 60611-3211. <http://www.acsnsqip.org>
7. Greenblatt, DY.; Rajamanickam, V.; Pugely, AJ.; and *et al*: Short-term outcomes after laparoscopic-assisted proctectomy for rectal cancer: results from the ACS NSQIP. *J Am Coll Surg*, 212(5): 844-54, 2011.
8. Ortega, G.; Rhee, DS.; Papandria, DJ.; Yang, J.; Ibrahim, AM.; Shore, AD.; Makary, MA.; and Abdullah, F. An Evaluation of Surgical Site Infections by Wound Classification System Using the ACS-NSQIP. *J Surg Res*, 2011.
9. Schoenfeld, AJ.; Ochoa, LM.; Bader, JO.; and Belmont, PJ., Jr. Risk factors for immediate postoperative complications and mortality following spine surgery. *J Bone Joint Surg Am*, 93(17): 1577-82, 2011.
10. Belmont, PJ., Jr.; Davey, S.; Orr, JD.; Ochoa, LM.; Bader, JO.; and Schoenfeld, AJ. Risk factors for 30-day postoperative complications and mortality after below-knee amputation. *J Am Coll Surg*, 213(3): 370-8, 2011.
11. SAS 9.3 for Windows. 2011. SAS Institute, Cary, NC.
12. Coca-Perrillon, M. Matching with Propensity Scores to Reduce Bias in Observational Studies. Proceedings of NESUG 2006. September 17-20, 2006, Philadelphia, Pennsylvania.
13. Lanehart, RE.; Rodriguez de Gil, P.; Kim, ES.; and *et al*. Propensity Score Analysis and Assessment of Propensity Score Approach Using SAS® PROCEDURES. Proceedings of SAS Global Forum. April 22-25, 2012. Orlando, Florida.
14. Parsons, LS. Using SAS® Software to Perform a Case-Control Match on Propensity Score in an Observational Study. Proceedings of SUGI 25. April 9-12, 2000, Indianapolis, Indiana.
15. Pugely, AJ, Martin, CT, Gao, Y, and Mendoza, SL. Outpatient Surgery Reduces Short-Term Complications in Lumbar Discectomy: An Analysis of 4,310 Patients from the ACS-NSQIP Database. *Spine (Phila Pa 1976)*. 38(3):264-271, February 01, 2013. DOI: 10.1097/BRS.0b013e3182697b57.

## CONTACT INFORMATION

Your comments and questions are valued and encouraged. Please contact the author at:

Yubo Gao  
 University of Iowa Hospitals and Clinics (UIHC)  
 Orthopaedic Surgery, 01066 JPP  
 200 Hawkins Dr.  
 Iowa City, IA 52242  
 Phone: 319-356-1674  
 Email: [yubo-gao@uiowa.edu](mailto:yubo-gao@uiowa.edu)

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.